# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-xvinfo
pkgver=1.1.3
pkgrel=2
pkgdesc="Prints out the capabilities of any video adaptors associated with the display that are accessible through the X-Video extension"
arch=('x86_64')
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libx11' 'libxv')
makedepends=('xorg-util-macros')
groups=('xorg-apps' 'xorg')
source=(https://xorg.freedesktop.org/archive/individual/app/xvinfo-${pkgver}.tar.bz2{,.sig})
sha512sums=('df8356898390d2aec4fbd15cee1627bb86b57ff8af41a527d4f6f6703a81f399a3098da342b94af6d4cf4a817d9fda1946a206bfbe3995e87ceb9910bb8b51bc'
            'SKIP')
validpgpkeys=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') #Alan Coopersmith <alan.coopersmith@oracle.com>

build() {
  cd xvinfo-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd xvinfo-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
